
Równania różniczkowe

Wstęp
Równania różniczkowe służą między innymi do opisu zjawisk zmiennych w czasie. W elektro-
technice opisują na przykład stany nieustalone powstające po przełączeniu obwodu. Rozwiązanie
analityczne bywa trudne lub niedostępne, dlatego stosuje się metody numeryczne, które wyzna-
czają przybliżone wartości rozwiązania w kolejnych punktach przedziału obliczeń [1, 3].

W ramach ćwiczenia omówione zostaną trzy metody wyznaczania rozwiązania równania
różniczkowego:

• wzór Taylora,

• metoda Eulera,

• metody Rungego-Kutty,

oraz ich zastosowanie do układów równań pierwszego rzędu i równań wyższych rzędów. Do
rozwiązania zagadnienia początkowego trzeba podać wartości początkowe, np. napięcie na kon-
densatorze i prąd cewki w chwili przełączenia. Krok ∆t wpływa na błąd dyskretyzacji i koszt
obliczeń. Jego zmniejszenie zwykle ogranicza błąd metody w zakresie stabilności, lecz po osią-
gnięciu pewnej skali korzyść może zostać zniwelowana przez błędy zaokrągleń.

W środowisku Scilab dostępna jest funkcja ode(). Rozwiązanie można również wyznaczyć
w programie Xcos, budując model z równań stanu [4].

Metody zostaną porównane na przykładzie napięcia kondensatora w szeregowym obwodzie
RC przyłączanym do źródła napięcia stałego E (ryc. 1).

Ryc. 1. Testowy obwód RC

Równanie różniczkowe dla analizowanego obwodu można wyprowadzić wychodząc z praw

Kirchhoffa, jak pokazano poniżej.

E = uR + uC (1)

i = C
duc

dt
−→ E = RC

duc

dt
+ uC (2)

duc

dt
= E − uc

RC
= E

RC
− uc

RC
(3)

Dla obwodu z ryc. 1 przyjęto R = 2000 Ω, E = 10 V i C = 25 µF. Przy warunku począt-
kowym uC(0) = 0 rozwiązanie analityczne równania (3) ma postać

uC = E − E · e− t
RC (4)

1 Wzór Taylora
Rozwinięcie Taylora pozwala wyznaczyć kolejną wartość rozwiązania, jeżeli można obliczyć jego
kolejne pochodne zwyczajne. Dla kroku h = ∆x i rozwinięcia rzędu m otrzymuje się [1, 2]:

y(x + ∆x) = y(x) +
m∑︂

n=1

1
n! · (∆x)n · y(n) (5)

gdzie m jest rzędem przybliżenia, a y(n) oznacza n-tą pochodną zwyczajną funkcji y.
Dla równania (3) druga pochodna wynosi

d2uC

dt2 = − 1
RC

duC

dt
= −E − uC

(RC)2 . (6)

W oparciu o algorytm (5) opracować można skrypt wyznaczający przebieg funkcji będącej
rozwiązaniem równania różniczkowego (3), jak pokazano poniżej.

clear;clc;clf;
R=2e3; E=10; C=25e-6; tau=R*C;
n=25; t0=0; tn=5*tau; dt=(tn-t0)/n;
t=[t0]; u=[0]; i=[0];
function u=rozwiazanie(t)

u=E*(1-exp(-t/tau));
endfunction
for k=1:n-1

du(1)=E/(R*C)-u(k)/(R*C);
du(2)=-du(1)/(R*C);
u(k+1)=u(k);
for j=1:2

u(k+1)=u(k+1)+(1/factorial(j))*(dt^j)*du(j);
end
t(k+1)=t(k)+dt;
i(k+1)=C*du(1)*1000;
plot2d(t,u,5);
plot2d(t,rozwiazanie(t),-4);

end
xlabel(’czas t [s]’)
ylabel(’Napięcie na kondensatorze uc [V]’)
os1=gca(); os2=newaxes();
plot2d(t,i,6);

os2.filled="off"; os2.axes_visible(1)="off";
os2.y_location="right";
ylabel(’Prąd w obwodzie i [mA]’)

Na ryc. 2 pokazano wyniki rozwinięcia pierwszego i drugiego rzędu dla 25 kroków (∆t =
0,01 s). Czerwona krzywa przedstawia rozwiązanie analityczne. Na ryc. 3 porównano wyniki po
zwiększeniu liczby kroków.

Ryc. 2. Rozwiązanie równania (3) wzorem Taylora

Ryc. 3. Zestawienie przebiegów dla 25 i 100 kroków obliczeń

Zwiększenie liczby kroków ograniczyło błąd dyskretyzacji. Wadą bezpośredniego rozwi-
nięcia Taylora jest konieczność wyznaczania kolejnych pochodnych rozwiązania; dla złożonych
równań może to być uciążliwe.

2 Metoda Eulera
Jawna metoda Eulera jest metodą pierwszego rzędu otrzymaną przez obcięcie rozwinięcia Tay-
lora po pierwszej pochodnej. Dla zagadnienia y′ = f(x, y) kolejne przybliżenie ma postać [1,
3]:

y(x + ∆x) = y(x) + ∆x · dy(x)
dx

(7)

Przykładowa implementacja ma postać:

clear;clc;clf;
R=2e3;E=10;C=25e-6; tau=R*C;
n=25; t0=0; tn=5*tau; dt=(tn-t0)/n;
t=[t0]; u=[0]; i=[0];
function u=rozwiazanie(t)

u=E*(1-exp(-t/tau));
endfunction
for k=1:n-1

du=E/(R*C)-u(k)/(R*C);
u(k+1)=u(k)+dt*du;
t(k+1)=t(k)+dt;
i(k+1)=C*du*1000;

end
plot2d(t,u,5);

plot2d(t,rozwiazanie(t),-4);
xlabel(’czas t [s]’)
ylabel(’Napięcie na kondensatorze uc [V]’)
os1=gca(); os2=newaxes();
plot2d(t,i,6);
os2.filled="off"; os2.axes_visible(1)="off";
os2.y_location="right";
ylabel(’Prąd w obwodzie i [mA]’)

Dla równania (3) otrzymano wyniki pokazane na ryc. 4 dla kroków ∆t = 0,01 s i ∆t =
0,001 s.

∆t = 0,01 s

∆t = 0,001 s

Ryc. 4. Rozwiązanie równania różniczkowego metodą Eulera

Można zauważyć, że dziesięciokrotne zmniejszenie kroku poprawiło zgodność z rozwiąza-
niem analitycznym (4). Ocena dokładności powinna jednak opierać się na błędzie liczbowym, a
nie wyłącznie na wizualnym pokrywaniu się krzywych.

3 Metody Rungego-Kutty
Metody Rungego-Kutty wyznaczają w jednym kroku kilka wartości prawej strony równania y′ =
f(x, y), nazywanych etapami. Odpowiednia kombinacja etapów pozwala uzyskać wyższy rząd
dokładności bez jawnego obliczania kolejnych pochodnych rozwiązania. Ogólna postać jawnej,
s-etapowej metody Rungego-Kutty jest następująca [1, 3]:

yn+1 = yn +
s∑︂

i=1
wiki, (8)

ki = h f

⎛⎝xn + cih, yn +
i−1∑︂
j=1

aijkj

⎞⎠ , i = 1, . . . , s. (9)

Współczynniki aij , ci i wi określają konkretną metodę. Liczba etapów s nie musi być równa
rzędowi metody. W tab. 1 zestawiono wybrane jawne metody rzędów od pierwszego do czwartego.

Tab. 1. Wybrane jawne metody Rungego-Kutty

Rząd wi ki Nazwa
I w1 = 1 k1 = ∆x · f(xn, yn) Eulera
II w1 = 1

2
w2 = 1

2

k1 = ∆x · f(xn, yn)
k2 = ∆x · f(xn + ∆x, yn + k1)

Heuna

III w1 = 1
6

w2 = 2
3

w3 = 1
6

k1 = ∆x · f(xn, yn)
k2 = ∆x · f(xn + 1

2∆x, yn + 1
2k1)

k3 = ∆x · f(xn + ∆x, yn − k1 + 2k2)

Kutta, rząd 3

IV w1 = 1
6

w2 = 1
3

w3 = 1
3

w4 = 1
6

k1 = ∆x · f(xn, yn)
k2 = ∆x · f(xn + 1

2∆x, yn + 1
2k1)

k3 = ∆x · f(xn + 1
2∆x, yn + 1

2k2)
k4 = ∆x · f(xn + ∆x, yn + k3)

klasyczna RK4

Klasyczna metoda RK4 jest popularnym kompromisem między kosztem jednego kroku a
dokładnością. Poniżej przedstawiono jej implementację dla równania (3).

clear;clc;clf;
R=2e3;E=10;C=25e-6; tau=R*C;
n=25; t0=0; tn=5*tau; dt=(tn-t0)/n;
t=[t0]; u=[0]; i=[0];
function u=rozwiazanie(t)

u=E*(1-exp(-t/tau));
endfunction
function y=rownanie(x)

y=E/(R*C)-x/(R*C);
endfunction
function [x,y]=RK4(dx,xw,yw)

k1=dx*rownanie(yw);
k2=dx*rownanie(yw+0.5*k1);
k3=dx*rownanie(yw+0.5*k2);
k4=dx*rownanie(yw+k3);
y=yw+(1/6)*k1+(1/3)*k2+(1/3)*k3+(1/6)*k4;
x=xw+dx;

endfunction
for k=1:n

[t(k+1),u(k+1)]=RK4(dt,t(k),u(k));

i(k+1)=C*rownanie(u(k))*1000;
end
plot2d(t,u,5);
plot2d(t,rozwiazanie(t),-4);
xlabel(’czas t [s]’)
ylabel(’Napięcie na kondensatorze uc [V]’)
os1=gca();
os2=newaxes();
plot2d(t,i,6);
os2.filled="off"; os2.axes_visible(1)="off";
os2.y_location="right";
ylabel(’Prąd w obwodzie i [mA]’)
legends(["uC" "i" "uC analityczne"],[5 6 -4],opt="lr");

Na ryc. 5 największy błąd ma metoda pierwszego rzędu. Różnice między metodami wyż-
szych rzędów są lepiej widoczne na powiększeniu (ryc. 6). Dziesięciokrotne zmniejszenie kroku
ogranicza różnice względem rozwiązania analitycznego, szczególnie dla metod wyższych rzędów
(ryc. 7 i 8).

Ryc. 5. Rozwiązanie równania metodami
Rungego-Kutty rzędów 1–4 dla ∆t = 0,01 s

Ryc. 6. Różnice pomiędzy rzędem 2-4 a przebie-
giem analitycznym

Ryc. 7. Rozwiązanie równania metodami
Rungego-Kutty rzędów 1–4 dla ∆t = 0,001 s

Ryc. 8. Różnice pomiędzy rzędem 3-4 a przebie-
giem analitycznym

4 Układ równań różniczkowych i równania wyższych rzędów
Opisane powyżej metody dedykowane są dla równań różniczkowych I rzędu. Obliczenia mogą być
realizowane także dla układu równań różniczkowych I rzędu. W przypadku równań lub układów
równań wyższych rzędów należy je sprowadzić do układu równań I rzędu.

Zastosowanie metod zostanie pokazane na przykładzie stanu nieustalonego w szeregowym
obwodzie RLC przyłączanym do źródła napięcia stałego (ryc. 9).

Ryc. 9. Testowy obwód RLC

Parametry testowego obwodu wynoszą: E = 100 V, R = 0,25 Ω, L = 25 mH i C = 33 mF.
Z drugiego prawa Kirchhoffa otrzymuje się

E = R · i + uL + uC (10)

Ponieważ prąd w obwodzie z kondensatorem jest opisany zależnością

i = C · duC

dt
(11)

a napięcie na cewce jest opisane zależnością

uL = L · di

dt
(12)

równanie napięciowe obwodu można zapisać

E = RC · duC

dt
+ L · di

dt
+ uC (13)

Zapisane równanie różniczkowe dwóch zmiennych można sprowadzić do równania róż-
niczkowego jednej zmiennej wyrażając natężenie prądu w obwodzie funkcją spadku napięcia na
kondensatorze.

E = RC · duC

dt
+ LC · d2uC

dt2 + uC (14)

Równanie drugiego rzędu (14) sprowadza się do układu dwóch równań pierwszego rzędu.
Po przekształceniu równań (10)–(12) otrzymuje się⎧⎪⎪⎨⎪⎪⎩

di

dt
= E − Ri − uC

L
duC

dt
= i

C

(15)

Wzór Taylora wymaga wyznaczenia pochodnych wyższych rzędów. Dla układu (15) otrzy-
muje się:

II.

⎧⎪⎪⎨⎪⎪⎩
d2i

dt2 = −
R di

dt + duC

dt

L
d2uC

dt2 = 1
C

· di

dt

(16)

III.

⎧⎪⎪⎨⎪⎪⎩
d3i

dt3 = −
R d2i

dt2 + d2uC

dt2

L
d3uC

dt3 = 1
C

· d2i

dt2

(17)

IV.

⎧⎪⎪⎨⎪⎪⎩
d4i

dt4 = −
R d3i

dt3 + d3uC

dt3

L
d4uC

dt4 = 1
C

· d3i

dt3

(18)

Błąd dyskretyzacji zależy od kroku i rzędu rozwinięcia. Na ryc. 10 pokazano wyniki dla
∆t = 0,01 s, a na ryc. 11 dla ∆t = 0,0001 s.

Ryc. 10. Rozwiązanie równania różniczkowego
wzorem Taylora dla rozwinięcia 1 i 4 dla ∆t =
0,01 s

Ryc. 11. Rozwiązanie równania różniczkowego
wzorem Taylora dla ∆t = 0,0001 s

Metody Rungego-Kutty stosuje się do układu w sposób analogiczny, przy czym każdy etap
ki jest wektorem o liczbie składowych równej liczbie równań.

5 Funkcja ode()
W środowisku Scilab funkcja ode() służy do numerycznego rozwiązywania równań i układów
równań różniczkowych. Wynik jest macierzą, której kolumny odpowiadają kolejnym wartościom
czasu [4].

Podstawowe wywołanie wymaga czterech argumentów:

y=ode(y0,t0,t,funkcja);

gdzie:

• y0 — wektor warunków początkowych,

• t0 — początkowa wartość zmiennej niezależnej,

• t — wektor punktów, w których ma zostać zwrócone rozwiązanie,

• funkcja — funkcja prawej strony układu równań stanu.

Jako zmienne stanu obwodu z ryc. 9 przyjęto{︄
x1 = uC ,

x2 = i.
(19)

Równania stanu mają zatem postać⎧⎪⎨⎪⎩
ẋ1 = x2

C
,

ẋ2 = E − Rx2 − x1
L

.
(20)

Kod skryptu obliczającego zdefiniowany równaniami stanu układ równań różniczkowych
dla obwodu z ryc. 9 pokazano poniżej wraz z uzyskanymi wynikami pokazanymi na ryc. 12.

clc; clear;clf;
function [y]=R_stanu(t,x)

y(1)=x(2)/C;
y(2)=(E-R*x(2)-x(1))/L;

endfunction
R=0.25; L=25e-3; C=33e-3; E=100;
dt=0.0001;
koniec=1;
t0=0; y0=[0;0];
t=t0:dt:koniec;
y=ode(y0,t0,t,R_stanu);
uc=y(1,:); i=y(2,:);
plot2d(t,[i’ uc’],[2 5]);
osie=gca(); osie.x_location="origin"; osie.y_location="origin";
xlabel("Czas t [s]");
legends(["i(t)" "u(t)"],[2 5], opt="ur");

Ryc. 12. Rozwiązanie równania różniczkowego funkcją ode()

Argument t musi występować w nagłówku funkcji prawej strony, nawet gdy układ jest
autonomiczny i czas nie pojawia się w równaniach jawnie.

6 Graficzne programowanie w XCOS

Xcos pozwala zbudować model układu równań różniczkowych w postaci schematu blokowego.
Dla rozpatrywanego obwodu model można oprzeć na równaniach (19)–(20). Stosuje się między
innymi bloki z grup:

• Systemy czasu ciągłego (INTEGRAL_m)

• Operacje matematyczne (GAIN_f, BIGSOM_f)

• Trasowanie sygnału (DEMUX_f)

• Źródła (CLOCK_c, CONST_f, GAINSIN_f,...)

• Sinks (CSCOPE lub CMSCOPE)

Przygotowując algorytm graficzny w programie należy zdefiniować warunki zadania (od-
noszące się do kroku czasowego i czasu trwania obliczeń) oraz można przedefiniować stałe (para-
metry obwodu) aby łatwiejsza była modyfikacja parametrów algorytmu. Konfiguracje tych pa-
rametrów realizuje się poprzez pozycje dostępne w zakładce symulacja, jak pokazano na ryc. 13.

Ryc. 13. Wybór okien konfiguracyjnych projektu

Ryc. 14. Ustawienie czasu trwania symulacji Ryc. 15. Definiowanie stałych

Równania różniczkowe opisujące analizowany układ z ryc. 9 można w postaci algorytmu
graficznego zaprojektować na dwa równoważne sposoby pokazane na ryc. 16 i ryc. 17.

Ryc. 16. Algorytm wersja A

Ryc. 17. Algorytm wersja B

Parametry bloków edytuje się po dwukrotnym kliknięciu elementu. Dla przebiegu symula-
cji istotna jest konfiguracja zegara i oscyloskopu. W zegarze należy ustawić okres próbkowania i
chwilę rozpoczęcia, np. 0,0001 i 0. W oscyloskopie definiuje się zakresy osi pionowych i osi czasu.

Zadania
Należy przeanalizować odpowiedź przejściową równoległego obwodu RLC zasilanego ideal-
nym źródłem prądowym is(t) = I0 1(t). Rezystor, cewka i kondensator są połączone równolegle,
a napięcie u(t) jest wspólne dla wszystkich gałęzi. Parametry i warunki początkowe należy przy-
jąć z poniższej tabeli zgodnie z numerem zestawu. Podczas laboratorium należy wyprowadzić
model, uruchomić obliczenia dla jednego wariantu i sprawdzić poprawność programu. Porówna-
nie kroków, klasyfikację odpowiedzi i bilans energetyczny należy dokończyć w sprawozdaniu.

Zestaw R [Ω] L [mH] C [µF] I0 [A] u(0) [V] iL(0) [A] tk [s]
1 8 100 100 1,0 0 0 0,15
2 15,81 100 100 1,0 0 0 0,15
3 40 100 100 1,0 0 0 0,15
4 10 80 120 0,8 0 0 0,14
5 18 120 82 1,2 0 0 0,18
6 32 90 150 1,0 1 0 0,16
7 12 150 68 1,5 0 0,1 0,20
8 25 70 100 0,6 0 0 0,12
9 50 110 47 1,0 0 0 0,12
10 14 95 110 0,9 -1 0 0,16
11 22 130 75 1,1 0 0,05 0,18
12 35 85 130 1,3 0 0 0,14
13 9 160 56 0,7 0 0 0,20
14 28 60 180 1,0 0,5 0 0,12
15 16 105 95 1,4 0 0 0,17

1. Narysować schemat obwodu, oznaczyć zwroty prądów gałęziowych i na podstawie pierw-
szego prawa Kirchhoffa wyprowadzić model stanu. Jako zmienne stanu przyjąć napięcie
obwodu u(t) oraz prąd cewki iL(t). Zapisać wspólną funkcję prawej strony układu równań;
gotowych równań obwodu szeregowego z przykładu nie należy przekształcać przez prostą
zamianę oznaczeń.

2. Zaimplementować jawną metodę Eulera i klasyczną metodę Rungego–Kutty czwartego
rzędu. Obie metody powinny korzystać z tego samego interfejsu danych i zwracać przebiegi
u(t), iL(t), iR(t) oraz iC(t).

3. Wyznaczyć rozwiązanie referencyjne funkcją ode(). Dla trzech wartości kroku 10−3 s,
2·10−4 s i 5·10−5 s porównać maksymalny błąd napięcia u(t) i prądu cewki iL(t), residuum
prądowego pierwszego prawa Kirchhoffa oraz czas obliczeń.

4. Z równania charakterystycznego modelu swobodnego wyprowadzić warunek tłumienia kry-
tycznego i obliczyć rezystancję Rkr. Dla wszystkich zestawów z tabeli rozpoznać odpowiedź
oscylacyjną, krytyczną albo aperiodyczną. Wyjaśnić, dlaczego zależność rodzaju odpowie-
dzi od rezystancji jest odwrotna niż w szeregowym obwodzie RLC omówionym w przykła-
dzie.

5. Dla jednego wariantu obliczyć energię zgromadzoną w cewce i kondensatorze oraz straty
mocy w rezystorze. Na podstawie przebiegów ocenić bilans mocy źródła, elementów ma-
gazynujących energię i rezystora.

Zadanie rozszerzające. Zastąpić skok prądu impulsem prostokątnym o czasie trwania 20 ms
albo przebiegiem sinusoidalnym o częstotliwości zbliżonej do częstotliwości własnej obwodu.
Porównać odpowiedź wymuszoną z odpowiedzią dla skoku prądu.

